Maple-assisted Proof of Empirical Formula for A269650

Robert Israel

18 Apr 2023

The arrays to be counted may be considered as directed walks on a directed pseudograph (i.e. a directed graph where loops are allowed). Since we only allow each of the pairs (0, 1) and (1,2) to occur at most once, each of 0, 1 and 2 will correspond to four nodes of this pseudograph, one for each answer to the pair of questions, has (0,1) occurred? has (1,2) occurred? The following code constructs the transition matrix, where the first four coordinates correspond to 0, the second four to 1 and the third four to 2.

```
> T:= Matrix(12,12):
  for i from 1 to 12 do T[i,i]:= 1 od: # loops
  T[1,6]:= 1: T[3,8]:= 1: # transitions 0 -> 1
  T[5,11]:= 1: T[6,12]:= 1: # transitions 1 -> 2
  for i from 1 to 4 do T[i,i+8]:= 1; T[i+4,i]:= 1; T[i+8,i]:= 1; T
  [i+8,i+4]:= 1 od:
  # transitions 0 -> 2, 1 -> 0, 2 -> 0 and 2 -> 1
> T;
```

$$\begin{bmatrix} 1 & 0 & 0 & 0 & 0 & 1 & 0 & 0 & 1 & 0 & 0 & 0 \\ 0 & 1 & 0 & 0 & 0 & 0 & 0 & 0 & 0 & 1 & 0 & 0 \\ 0 & 0 & 1 & 0 & 0 & 0 & 0 & 1 & 0 & 0 & 1 & 0 \\ 0 & 0 & 0 & 1 & 0 & 0 & 0 & 0 & 0 & 0 & 0 & 1 \\ 1 & 0 & 0 & 0 & 1 & 0 & 0 & 0 & 0 & 0 & 1 & 0 \\ 0 & 1 & 0 & 0 & 0 & 1 & 0 & 0 & 0 & 0 & 0 & 1 \\ 0 & 0 & 1 & 0 & 0 & 0 & 1 & 0 & 0 & 0 & 0 & 0 \\ 0 & 0 & 0 & 1 & 0 & 0 & 0 & 1 & 0 & 0 & 0 & 0 \\ 1 & 0 & 0 & 0 & 1 & 0 & 0 & 0 & 1 & 0 & 0 & 0 \\ 0 & 1 & 0 & 0 & 0 & 1 & 0 & 0 & 0 & 1 & 0 & 0 \\ 0 & 0 & 1 & 0 & 0 & 0 & 1 & 0 & 0 & 0 & 1 & 0 \\ 0 & 0 & 0 & 1 & 0 & 0 & 0 & 1 & 0 & 0 & 0 & 1 \end{bmatrix} \quad \textbf{(1)}$$

Then $a(n) = u^T T^{n-1} v$ where $u^T = [1, 0, 0, 0, 1, 0, 0, 0, 1, 0, 0, 0]$ (corresponding to the fact that we start with neither 0-1 nor 1-2 having occurred), and $v^T = [1.1.1.1.1.1.1.1.1.1.1.1]$. To check, here are the first few terms in the sequence.

```
> u:= <1,0,0,0,1,0,0,0,1,0,0,0>: v:= <1$12>:
  seq(u^%T . T^i . v, i = 0 .. 20);
```

3, 9, 27, 79, 225, 626, 1710, 4605, 12259, 32320, 84504, 219356, 565816, 1451349, 3704271, **(2)** 9412153, 23818707, 60055275, 150913073, 378064818, 944442242

This must satisfy a linear recurrence, corresponding to a polynomial dividing the minimal polynomial of the matrix T. In this case, it actually is the minimal polynomial.

```
> P:= LinearAlgebra:-MinimalPolynomial(T,x);
```

$$P := x^9 - 9\,x^8 + 33\,x^7 - 66\,x^6 + 84\,x^5 - 75\,x^4 + 47\,x^3 - 21\,x^2 + 6\,x - 1 \tag{3}$$

Thus the recurrence is

```
> add(coeff(P,x,i)*a(n+i-9),i=0..9) = 0;
```

$$-a(n-9) + 6\,a(n-8) - 21\,a(n-7) + 47\,a(n-6) - 75\,a(n-5) + 84\,a(n-4) - 66\,a(n-3) + 33\,a(n-2) - 9\,a(n-1) + a(n) = 0 \tag{4}$$